Problem:
Score : 400 points

Problem Statement

We have a large square grid with H rows and W columns.
Iroha is now standing in the top-left cell.
She will repeat going right or down to the adjacent cell, until she reaches the bottom-right cell.

However, she cannot enter the cells in the intersection of the bottom A rows and the leftmost B columns. (That is, there are A×B forbidden cells.) There is no restriction on entering the other cells.

Find the number of ways she can travel to the bottom-right cell.

Since this number can be extremely large, print the number modulo 10^9+7.

Constraints

1 ≦ H, W ≦ 100,000

1 ≦ A < H

1 ≦ B < W

Input

The input is given from Standard Input in the following format:

H W A B

Output

Print the number of ways she can travel to the bottom-right cell, modulo 10^9+7.

Sample Input 1

2 3 1 1

Sample Output 1

2

We have a 2×3 grid, but entering the bottom-left cell is forbidden. The number of ways to travel is two: "Right, Right, Down" and "Right, Down, Right".

Sample Input 2

10 7 3 4

Sample Output 2

3570

There are 12 forbidden cells.

Sample Input 3

100000 100000 99999 99999

Sample Output 3

1

Sample Input 4

100000 100000 44444 55555

Sample Output 4

738162020

Solution in OCaml (4.02.3) (judge verdict: Accepted)

let modulo = 1000000007
let (+@) a b = (a + b) mod modulo
let ( *@ ) a b = (a * b) mod modulo

let rec mod_pow a n =
  if n <= 0 then 1
  else if n mod 2 = 1 then a *@ mod_pow a (n-1)
  else let h = mod_pow a (n/2) in h *@ h
let mod_inv a = mod_pow a (modulo-2)

let lim = 200010

let fact =
  let res = Array.make lim 1 in
  for i = 1 to lim-1 do res.(i) <- i *@ res.(i-1) done;
  res

let inv_fact = Array.map mod_inv fact

let comb n k =
  if n < k then 0 else fact.(n) *@ inv_fact.(k) *@ inv_fact.(n-k)

let r h w = comb (h+w) h

let () =
  Scanf.scanf "%d %d %d %d" @@ fun h w a b ->
  Array.init (h-a) (fun i -> r i (b-1) *@ r (h-1-i) (w-b-1))
  |> Array.fold_left (+@) 0
  |> Printf.printf "%d\n"